Pensieve Header: Wheeled Semi-Symmetrized calculus in the 2D quotient.

Continues “Semi-Symmetrized 2D Calculus.nb”, “A 2D B-Picture, IV.nb”, uses computations from “BCH in Blobs.nb”.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\w-Computations"]
```

```
C:\drorbn\AcademicPensieve\Projects\w-Computations
```

```
ar[i_, j_] := t[i] h[j]
```

```
μCollect[μ_] := Collect[μ, _h, Collect[#, _t, FullSimplify] &];
SetAttributes[μForm, Listable];
μForm[μ_] := Module[
  {tails, heads, mat},
  tails = Union[Cases[μ, t[s_] :→ s, Infinity]];
  heads = Union[Cases[μ, h[s_] :→ s, Infinity]];
  mat = Outer[FullSimplify[Coefficient[μ, h[#1] t[#2]]] &, heads, tails];
  PrependTo[mat, t /@ tails];
  mat = Prepend[Transpose[mat],
    Prepend[h /@ heads, μ /. (h[_] | t[_]) → 0]
   ];
  MatrixForm[mat]
 ]
```

Heads Works

```
hm[x_, y_, z_][μ_] := Module[
   {ξ, η},
   ξ = D[μ, h[x]];
   η = D[μ, h[y]];
   μCollect[(μ /. h[x | y] → 0) + ξ h[z] + (1 + ξ /. t[s_] :→ c[s]) η h[z]]
  ];
hΔ[z_, x_, y_][μ_] := μCollect[μ /. h[z] → h[x] + h[y]];
hS[x_][μ_] := Module[{β},
   β = 1 + D[μ, h[x]] /. t[s_] :→ c[s];
   μCollect[μ /. h[x] → -h[x] / β]
  ];
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]]
```

```
h[5] (t[1] + (1 + c[1]) t[2])
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // μForm
```

$$\begin{pmatrix} 0 & h[5] \\ t[1] & 1 \\ t[2] & 1+c[1] \end{pmatrix}$$

Associativity of Heads Multiplication

```
μ1 = α1 ar[1, 1] + α2 ar[2, 2] + α3 ar[3, 3]
```

```
α1 h[1] t[1] + α2 h[2] t[2] + α3 h[3] t[3]
```

```
μ1 // μForm
```

0	h[1]	h[2]	h[3]
t[1]	α1	0	0
t[2]	0	α2	0
t[3]	0	0	α3

```
μ1 // hm[1, 2, 1] // μForm
```

0	h[1]	h[3]
t[1]	α1	0
t[2]	α2 + α1 α2 c[1]	0
t[3]	0	α3

```
(t1 = μ1 // hm[1, 2, 1] // hm[1, 3, 1]) // μForm
```

0	h[1]
t[1]	α1
t[2]	α2 + α1 α2 c[1]
t[3]	α3 (1 + α1 c[1]) (1 + α2 c[2])

```
(t2 = μ1 // hm[2, 3, 2] // hm[1, 2, 1]) // μForm
```

0	h[1]
t[1]	α1
t[2]	α2 + α1 α2 c[1]
t[3]	(1 + α1 c[1]) (α3 + α2 α3 c[2])

```
(t1 - t2 ) // μForm
```

0	h[1]
t[1]	0
t[2]	0
t[3]	0

▪ Compatibility of m and Δ

```
μ = α1 ar[1, 1] + α2 ar[2, 2];
{
  μ // hΔ[1, 1, 3] // hΔ[2, 2, 4] // hm[1, 2, 1] // hm[3, 4, 2],
  μ // hm[1, 2, 1] // hΔ[1, 1, 2]
 } // μForm
```

{

0	h[1]	h[2]
t[1]	α1	α1
t[2]	α2 + α1 α2 c[1]	α2 + α1 α2 c[1]

,

0	h[1]	h[2]
t[1]	α1	α1
t[2]	α2 + α1 α2 c[1]	α2 + α1 α2 c[1]

}

▪ The Antipode Property

```
{
 α ar[1, 1] // hΔ[1, 1, 2] // hS[2] // hm[1, 2, 1],
 α ar[1, 1] // hΔ[1, 1, 2] // hS[2] // hm[2, 1, 1]
}
```

{0, 0}

Factorization

```
hfac[z_, xtails_List → x_, y_][μ_] := Module[
   {ytails},
   ytails = Complement[
     Union[Cases[μ, t[s_] ⧴ s, Infinity]],
     xtails
    ];
   hfac[z, xtails → x, ytails → y][μ]
  ];
hfac[z_, x_, ytails_List → y_][μ_] := Module[
   {xtails},
   xtails = Complement[
     Union[Cases[μ, t[s_] ⧴ s, Infinity]],
     ytails
    ];
   hfac[z, xtails → x, ytails → y][μ]
  ];
hfac[z_, xtails_List → x_, ytails_List → y_][μ_] := Module[
  {ζ, ξ, η},
  ζ = D[μ, h[z]];
  ξ = ζ /. ((t[#] → 0) & /@ ytails);
  η = ζ /. ((t[#] → 0) & /@ xtails);
  μCollect[μ - h[z] ζ + h[x] ξ + h[y] η / (1 + ξ /. t[s_] → c[s])]
 ]
```

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // μForm
```

0	h[5]
t[1]	1
t[2]	1 + c[1]

```
hm[3, 4, 5][ar[1, 3] + ar[2, 4]] // hfac[5, {1} → 3, 4]
```

h[3] t[1] + h[4] t[2]

Conjugation

```
(μ2 = α1 ar[1, 1] + α2 ar[2, 1] + α3 ar[2, 2] + α4 ar[2, 3]) // μForm
```

0	h[1]	h[2]	h[3]
t[1]	α_1	0	0
t[2]	α_2	α_3	α_4

Warning: Presently conj is designed to work only in the no-feedback case.

```
conj[y_, x_][μ_] := Module[
   {ξ, η, a},
   ξ = D[μ, h[x]];
   η = D[μ, t[y]];
   a = 1 + ξ /. t[s_] ⧴ c[s];
   μCollect[(μ /. t[y] → a t[y]) - c[y] ξ η]
  ];
```

```
μ2 // conj[1, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 (1 + c[2]\,\alpha_3) & 0 & 0 \\ t[2] & \alpha_2 - c[1]\,\alpha_1\,\alpha_3 & \alpha_3 & \alpha_4 \end{pmatrix}$$

```
(t1 = μ2 // conj[1, 2] // conj[1, 3] // hm[2, 3, 2]) // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 (1 + c[2]\,\alpha_3)(1 + c[2]\,\alpha_4) & 0 \\ t[2] & \alpha_2 - c[1]\,\alpha_1 (\alpha_4 + \alpha_3 (1 + c[2]\,\alpha_4)) & \alpha_4 + \alpha_3 (1 + c[2]\,\alpha_4) \end{pmatrix}$$

```
(t2 = μ2 //  hm[2, 3, 2] // conj[1, 2]) // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 (1 + c[2]\,\alpha_3)(1 + c[2]\,\alpha_4) & 0 \\ t[2] & \alpha_2 - c[1]\,\alpha_1 (\alpha_4 + \alpha_3 (1 + c[2]\,\alpha_4)) & \alpha_4 + \alpha_3 (1 + c[2]\,\alpha_4) \end{pmatrix}$$

```
t1 == t2
```

True

■ "4T"

```
Riffle[
 ComposeList[
   ops = {conj[2, 1], h∆[1, 1, 3], hm[2, 3, 2], h∆[1, 1, 3], hS[3], hm[3, 2, 2]},
   α1 ar[1, 1] + α2 ar[2, 2]
  ] // μForm,
 ops
]
```

$\left\{ \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & 0 \\ t[2] & 0 & \alpha_2 \end{pmatrix}, \text{conj}[2, 1], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & -c[2]\,\alpha_1\,\alpha_2 \\ t[2] & 0 & (1 + c[1]\,\alpha_1)\,\alpha_2 \end{pmatrix}, \right.$

$\text{h}\triangle[1, 1, 3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & -c[2]\,\alpha_1\,\alpha_2 & \alpha_1 \\ t[2] & 0 & (1 + c[1]\,\alpha_1)\,\alpha_2 & 0 \end{pmatrix}, \text{hm}[2, 3, 2],$

$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_1 \\ t[2] & 0 & (1 + c[1]\,\alpha_1)\,\alpha_2 \end{pmatrix}, \text{h}\triangle[1, 1, 3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_1 & \alpha_1 \\ t[2] & 0 & (1 + c[1]\,\alpha_1)\,\alpha_2 & 0 \end{pmatrix},$

$\left. \text{hS}[3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_1 & -\frac{\alpha_1}{1 + c[1]\,\alpha_1} \\ t[2] & 0 & (1 + c[1]\,\alpha_1)\,\alpha_2 & 0 \end{pmatrix}, \text{hm}[3, 2, 2], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & 0 \\ t[2] & 0 & \alpha_2 \end{pmatrix} \right\}$

```
α1 ar[1, 1] + α2 ar[2, 2] (Exp[c[2]] - 1) / c[2] // conj[2, 1] // h∆[1, 1, 3] //
    hm[2, 3, 2] // h∆[1, 1, 3] // hS[3] // hm[3, 2, 2] // μForm
```

$$\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & 0 \\ t[2] & 0 & \frac{(-1 + e^{c[2]})\,\alpha_2}{c[2]} \end{pmatrix}$$

```
Riffle[
 ComposeList[
   ops = {hΔ[1, 1, 3], hm[2, 3, 2], hΔ[1, 1, 3], hS[3], hm[3, 2, 2]},
   α1 ar[1, 1] + α2 ar[1, 2]
  ] // μForm,
 ops
]
```

$$\left\{\begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_2 \end{pmatrix}, h\triangle[1, 1, 3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_2 & \alpha_1 \end{pmatrix}, hm[2, 3, 2], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_2 + \alpha_1 (1 + c[1]\,\alpha_2) \end{pmatrix}, h\triangle[1, 1, 3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_2 + \alpha_1 (1 + c[1]\,\alpha_2) & \alpha_1 \end{pmatrix}, hS[3], \begin{pmatrix} 0 & h[1] & h[2] & h[3] \\ t[1] & \alpha_1 & \alpha_2 + \alpha_1 (1 + c[1]\,\alpha_2) & -\frac{\alpha_1}{1 + c[1]\,\alpha_1} \end{pmatrix}, hm[3, 2, 2], \begin{pmatrix} 0 & h[1] & h[2] \\ t[1] & \alpha_1 & \alpha_2 \end{pmatrix}\right\}$$